

Single Factor Experiments with Repeated Measures

In a wine-judging competition, four Chardonnay wines of the same vintage were judged by six experienced judges. Each judge tasted the wines in a blind fashion, i.e. without knowing their identities. The order of the wine presentation was randomized independently for each judge. To reduce carryover and other interference effects, the judges did not drink the wines and rinsed their mouths thoroughly between tastings. Each wine was scored on a 40-point scale; the higher the score, the greater is the excellence of the wine. The six judges are considered to be a random sample from the population of possible judges, while the four wines tasted of interest in themselves.

```

titlel " ";
options nodate nonumber ls=80 nocenter;
data wine;
input rating judge wine @@;
cards;
20 1 1 24 1 2 28 1 3 28 1 4 15 2 1 18 2 2
23 2 3 24 2 4 18 3 1 19 3 2 24 3 3 23 3 4
26 4 1 26 4 2 30 4 3 30 4 4 22 5 1 24 5 2
28 5 3 26 5 4 19 6 1 21 6 2 27 6 3 25 6 4
;

```

```

symbol1 i=j line=1 v=plus c=black;
symbol2 i=j line=5 v=square c=black;
symbol3 i=j line=10 v=triangle c=black;
symbol4 i=j line=15 v=star c=black;
symbol5 i=j line=20 v=dot c=black;
symbol6 i=j line=25 v=point c=black;

```

```

proc gplot;
plot rating*wine=judge;
run;

```

```

proc glm;
class judge wine;
model rating = judge wine;
random judge / test;
lsmeans wine / pdiff cl adj=tukey;
output out=residata p=yhat
rstudent=stdres;
run;

```

```

proc univariate plot;
var stdres; run;

```

```


symbol1 v=plus c=black; symbol2 v=square c=black; symbol3 v=triangle c=black;
symbol4 v=star c=black; symbol5 v=dot c=black; symbol6 v=point c=black;

```

```

proc gplot;
plot stdres*yhat=judge;
run;

```



Source	Type III Expected Mean Square
judge	Var(Error) + 4 Var(judge)
wine	Var(Error) + Q(wine)

Tests of Hypotheses for Mixed Model Analysis of Variance

Dependent Variable: rating


Source	DF	Type III SS	Mean Square	F Value	Pr > F
judge	5	173.333333	34.666667	32.50	<.0001
wine	3	184.000000	61.333333	57.50	<.0001
Error: MS(Error)	15	16.000000	1.066667		

Adjustment for Multiple Comparisons: Tukey

wine	rating LSMEAN	LSMEAN Number
1	20.0000000	1
2	22.0000000	2
3	26.6666667	3
4	26.0000000	4

Least Squares Means for Effect wine

i	j	Difference Between Means	Simultaneous 95% Confidence Limits for LSMean(i) - LSMean(j)
1	2	-2.000000	-3.718582 -0.281418
1	3	-6.666667	-8.385248 -4.948085
1	4	-6.000000	-7.718582 -4.281418
2	3	-4.666667	-6.385248 -2.948085
2	4	-4.000000	-5.718582 -2.281418
3	4	0.666667	-1.051915 2.385248



Wines 3 and 4 are judged best, and do not differ significantly from each other. Wines 1 and 2 are judged to be inferior to wines 3 and 4, with wine 1 receiving a mean rating significantly lower than wine 2.

Stem Leaf	#	Boxplot
2 33	2	
1		
1 2	1	
0 88888	5	+-----+
0 0000	4	+
-0 44444444	8	*-----*
-0		
-1 2	1	
-1 777	3	
-----+-----+-----+-----+		

